

Fun with the SAS® FILENAME Statement



Email: Mark.Jordan@sas.com
Twitter: [@SASJedi](https://twitter.com/SASJedi)
Blog: <http://sasjedi.tips>
Author: <http://support.sas.com/jordan>




Introduction – Topics we will cover

- Without first downloading the file:
 - Open and read raw data files located on a remote FTP server
 - Open and read a text file located on a remote web server
- PROC HTTP as a quick way to download data files from the internet
- Read and write to files stored in a ZIP archive
- Execute operating system commands and process the results like a text file
- Have your SAS program send email on your behalf

Retrieving Data via FTP

Let's get started retrieving and importing WWW data, starting with data available from an FTP server:




3

Copyright © SAS Institute Inc. All rights reserved.



Importing Delimited Text Files

The donors.csv file is in the /free_data directory:



I'd like to use this data in my SAS program. Let's get it using FILENAME FTP.

4

Copyright © SAS Institute Inc. All rights reserved.




Retrieving a Delimited Text file via FTP

FILENAME FTP allows accessing the data in place

Syntax:

FILENAME fileref FTP 'external-file' <ftp options>;

Example:



FWF01-filename_ftp.sas



5

Copyright © SAS Institute Inc. All rights reserved.

Retrieving Data: FTP

PROC IMPORT will write a data step for us, which can be viewed in the SAS Log, or retrieved into the Enhanced Editor using the RECALL command:

```
data WORK.DONORS_CSV;
  %let _EFIERR_ = 0; /*set ERROR detection macro variable*/
  infile MYFTP delimiter = ',' MISSOVER DSD firstobs=2;
  informat Name $21.;
  informat Donation best32.;
  informat Recipients $59.;
  format Name $21.;
  format Donation best12.;
  format Recipients $59.;
  input Name $ Donation Recipients $;
  if _ERROR_ then call symputx('_EFIERR_',1);
  /*set ERROR detection macro variable*/
run;
```

6

Copyright © SAS Institute Inc. All rights reserved.





Importing Delimited Text via FTP

This demonstration illustrates importing delimited text data into SAS using FILENAME FTP and PROC IMPORT, without first downloading the text file from the FTP server.

FWF01-filename_ftp.sas



Copyright © SAS Institute Inc. All rights reserved.

Retrieving Data: FTP

A quick query to check the data...

```
proc FedSQL;  
title "&SYSLAST";  
title2 "Imported from CSV using FILENAME FTP and DATA step";  
select *  
  from WORK.Donors_csv limit 5;  
quit;
```

FWF01-filename_ftp.sas



8

Copyright © SAS Institute Inc. All rights reserved.

Retrieving Data: FTP

Partial Output

WORK.DONORS_CSV		
Imported from CSV using FILENAME FTP and DATA step		
Name	Donation	Recipients
Ray Abbott	40	Vox Victimas
Tameaka Akinfolarin	40	EarthSalvors 40%, Vox Victimas 60%
Nishan Apr	40	Child Survivors
Jeryl Areu	40	Mitleid International 30%, Save the Baby Animals 70%
Debra Amant	60	Cancer Cures, Inc.

That worked well! Let's try the same trick with the Donors.xlsx Excel file...


9

Copyright © SAS Institute Inc. All rights reserved.



Importing Excel data from the Internet

I like to access Excel files using either the SAS/Access to PC File Formats engines using either LIBNAME statement or PROC IMPORT.



The screenshot shows an Excel spreadsheet titled 'donors.xlsx - Excel'. The spreadsheet has three columns: Name, Donation, and Recipients. The data is as follows:

Name	Donation	Recipients
Ray Abbott	\$40.00	Vox Victimas
Tameaka Akinfolarin	\$40.00	EarthSalvors 40%, Vox Victimas 60%
Nishan Apr	\$40.00	Child Survivors
Jeryl Areu	\$40.00	Mitleid International 30%, Save the Baby Animals 70%
Debra Amant	\$60.00	Cancer Cures, Inc.
John Atkins	\$40.00	Cancer Cures, Inc. 60%, Cuidadores Ltd. 40%
Wanda Branly	\$25.00	Mitleid International 90%, Save the Baby Animals 10%
Saunders Braggi	\$40.00	Conserve Nature, Inc. 40%, AquaMissions International 60%
Carmelo Broome	\$10.00	Child Survivors 10%, Disaster Assist, Inc. 90%

10

Copyright © SAS Institute Inc. All rights reserved.



Retrieving Data: FTP

Let's try accessing the Excel file in place like we did with the CSV file:

```
filename MyFTP ftp "donors.xlsx" cd="/free_data"
        user="anonymous" pass="myname@somewhere.com"
        host="ftp.mydomian.net" debug;
proc import datafile=MyFTP out=WORK.Donors_xls
        dbms=XLSX replace;
run;
```

Partial SAS log:

```
ERROR: This "filename FTP" access method is not
supported by "proc import". Please copy the file
to local disk before running the procedure.
```

FWF01-filename_ftp.sas



11

Copyright © SAS Institute Inc. All rights reserved.

Retrieving Data: FTP

What went wrong?

- To leverage the Excel engine a PHYSICAL file location is required - a FILEREF won't do.
- If we can't use a FILEREF, FILENAME FTP won't work.
- We'll need to copy the file to local disk
(as suggested by the SAS Log ERROR message)

...so let's use SAS and PROC HTTP to drive the downloading process!

12

Copyright © SAS Institute Inc. All rights reserved.





Retrieving Data: PROC HTTP

```
filename xl "&path\donors.xlsx";  
proc http  
  url="http://ftp.mydomian.net/free_data/donors.xlsx"  
  out=xl;  
run;
```

13

Copyright © SAS Institute Inc. All rights reserved.



Retrieving Data: FTP

In the PROC IMPORT step, point to the local copy of the file:

```
proc import datafile="&path\donors.xlsx"  
  out=WORK.Donors_xlsx  
  dbms=XLSX replace;  
run;  
title "Imported from donors.xlsx";  
proc print data=donors_xlsx(obs=10);  
run;
```

14

Copyright © SAS Institute Inc. All rights reserved.

FWF02-proc_http_excel.sas





Importing Excel Data via FTP

This demonstration illustrates downloading the Excel file from the FTP server using PROC HTTP.

FWF02-proc_http_excel.sas



Copyright © SAS Institute Inc. All rights reserved.

Retrieving Data: FTP

Partial SAS Log:

```
NOTE: 8 records were written to the file
      "C:\UntangledWeb\Downloads\script.ftp".
NOTE: File downloaded successfully in 1 seconds.
NOTE: DATA statement used (Total process time):
      real time          0.35 seconds
      cpu time           0.20 seconds
NOTE: WORK.MYDATA_XLS data set was successfully created.
```

Partial Output:

Imported from Excel after FTP Download (5 rows)			
Row	Name	Donation	Recipients
1	Ray Abbott	\$40.00	Vox Victimias
2	Tameaka Akinfolarin	\$40.00	EarthSalvors 40%, Vox Victimias 60%
3	Nishan Apr	\$40.00	Child Survivors
4	Jeryl Areu	\$40.00	Mitleid International 30%, Save the Baby Animals 70%
5	Debra Armant	\$60.00	Cancer Cures, Inc.


Copyright © SAS Institute Inc. All rights reserved.



Retrieving Data: URL

What if there is no FTP server?

Can we get data files with an "http://" URL instead?



17

Copyright © SAS Institute Inc. All rights reserved.



Retrieving Data: URL

We can access the file in place using FILENAME URL

Syntax:

FILENAME fileref URL 'external-file' <url-options>;

Example:

```
filename MyURL URL
"http://mydomian.net/free_data/donors.csv"
/*user="UserName" pass="myname@somewhere.com"*/;
```

FWF03-filename_url.sas

18

Copyright © SAS Institute Inc. All rights reserved.





Importing Delimited Text via FILENAME URL

This demonstration illustrates importing delimited text data from a web server using FILENAME URL and PROC IMPORT.

FWF03-filename_url.sas



Copyright © SAS Institute Inc. All rights reserved.

Retrieving Data: URL

Partial SAS Log:

```
NOTE: 125 records were read from the infile MYFTP.  
      The minimum record length was 0.  
      The maximum record length was 155.
```

Partial Output:

Imported from CSV using FILENAME URL (5 obs)			
Row	Name	Donation	Recipients
1	Ray Abbott	40	Vox Victimas
2	Tameaka Akinfolarin	40	EarthSalvors 40%, Vox Victimas 60%
3	Nishan Apr	40	Child Survivors
4	Jeryl Areu	40	Mitleid International 30%, Save the Baby Animals 70%
5	Debra Armant	60	Cancer Cures, Inc.

That worked well!

20

Copyright © SAS Institute Inc. All rights reserved.



Retrieving Excel Data: URL

Partial SAS Log:

```
ERROR: This "filename URL" access method is not  
supported by "proc import". Please copy the file  
to local disk before running the procedure.
```

Just as with FILENAME FTP, this technique won't work to import Excel files.

www03-filename_url.sas



21

Copyright © SAS Institute Inc. All rights reserved.

Retrieving Excel Data: URL

Once again, we can use PROC HTTP to download the file:

```
filename resp "s:\workshop\donors.xlsx";  
  
proc http  
  url="http://mydomian.net/free_data/donors.xlsx"  
  out=resp;  
run;
```

www04-call_system_url.sas



22

Copyright © SAS Institute Inc. All rights reserved.

Retrieving Excel Data: URL

Once downloaded, we can finish the job with the same PROC IMPORT code we previously used for Excel:

```
proc import datafile=resp  
  out=Work.Donors_xls_url  
  dbms=xlsx replace;  
run;
```

www04-call_system_url.sas



23

Copyright © SAS Institute Inc. All rights reserved.



Importing Excel Data via HTTP

This demonstration illustrates by downloading the Excel file via HTTP using PROC HTTP.

www04-call_system_URL.sas



Copyright © SAS Institute Inc. All rights reserved.

Retrieving Data: URL

Partial Output:

Imported from EXCEL downloaded via HTTP			
Row	Name	Donation	Recipients
1	Ray Abbott	\$40.00	Vox Victimas
2	Tameaka Akinfolarin	\$40.00	EarthSalvors 40%, Vox Victimas 60%
3	Nishan Apr	\$40.00	Child Survivors
4	Jeryl Areu	\$40.00	Mitleid International 30%, Save the Baby Animals 70%
5	Debra Armant	\$60.00	Cancer Cures, Inc.

25

Copyright © SAS Institute Inc. All rights reserved.



Working with ZIP Archives

DOWNLOAD a ZIP file using PROC HTTP

```
filename myZIP "&path\donors.zip";
proc http
  url="http://mydomian.net/free_data/donors.zip"
  out=myZIP;
run;
```

FWF05-DonorZipDownload.sas


26

Copyright © SAS Institute Inc. All rights reserved.



Working with ZIP Archives

[FILENAME ZIP](#) makes it easy to work with ZIP files programmatically in SAS



27

Copyright © SAS Institute Inc. All rights reserved.



Working with ZIP Archives

Assign a FILEREF to the zip file using the ZIP protocol:

```
FILENAME MyZIP ZIP "&path/donors.zip";
```

Reference individual members of the archive in your INFILE statement:

```
data WORK.Donors zip;
  infile MyZip(donors.txt) delimiter='09'x
    MISSOVER DSD firstobs=2;
  input Employee_ID Employee_Name:$25. Street_Number
    Street_Name:$20. City:$15. State:$2. Postal_Code
    Country:$2. Qtr1:comma. Qtr2:comma. Qtr3:comma.
    Qtr4:comma. AnnualDonation:comma. Recipients:$75.;
  format Qtr1-Qtr4 AnnualDonation comma12.;
run;
```

FWF06-filename_zip.sas

28

Copyright © SAS Institute Inc. All rights reserved.



Working with ZIP Archives

Partial SAS Log:

```
NOTE: The infile library MYZIP is:
      Directory=C:\UntangledWeb\Downloads\donors.zip

NOTE: The infile MYZIP(donors.txt) is:
      Filename=C:\UntangledWeb\Downloads\donors.zip,
      Member Name=donors.txt

NOTE: A total of 124 records were read from the infile library MYZIP.
      The minimum record length was 131.
      The maximum record length was 185.
NOTE: 124 records were read from the infile MYZIP(donors.txt).
      The minimum record length was 131.
      The maximum record length was 185.
NOTE: The data set WORK.DONORS_ZIP has 124 observations and 14 variables.
```

29

Copyright © SAS Institute Inc. All rights reserved.



Working with ZIP Archives

Partial Output:

Imported from ZIP file member donors.txt			
Country	Employee_Name	AnnualDonation	Recipients
US	Abbott, Ray	40	Vox Victimas
US	Akinfolarin, Tameaka	40	EarthSalvors 40%, Vox Victimas 60%
US	Apr, Nishan	40	Child Survivors
US	Areu, Jeryl	40	Mitleid International 30%, Save the Baby Animals 70%
US	Amant, Debra	60	Cancer Cures, Inc.
US	Atkins, John	40	Cancer Cures, Inc. 60%, Cuidadores Ltd. 40%
US	Branly, Wanda	25	Mitleid International 90%, Save the Baby Animals 10%
US	Briggi, Saunders	40	Conserve Nature, Inc. 40%, AquaMissions International 60%
US	Broome, Camelo	10	Child Survivors 10%, Disaster Assist, Inc. 90%
US	Buckner, Bumetta	20	Disaster Assist, Inc. 50%, Cancer Cures, Inc. 50%

30

Copyright © SAS Institute Inc. All rights reserved.





Working with ZIP Archives

If the ZIP file contains folder information, you can access a member in a ZIPPED folder like this:

```
infile MyZip(folder/filename.ext);
```

31

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL

First, make sure your SAS System options are properly set for sending email

```
options EMAILSYS=SMTP EMAILID="mark.jordan@sas.com" EMAILPW="&EMAILPW"
      EMAILHOST="&EMAILHOST" EMAILPORT=25;

filename mymail email "mark.jordan@sas.com" subject="This is a test";

title;
data _null_;
  file mymail;
  put 'Emailing from SAS is fun!';
  put 'Stay SASy!';
  put 'Mark';
run;
```

FWF07-sas_email_options.sas

32

Copyright © SAS Institute Inc. All rights reserved.





Setting and testing SAS email options


This demonstration illustrates setting and testing SAS email options.

FWF07-sas_email_options.sas



Copyright © SAS Institute Inc. All rights reserved.

Emailing from SAS® with FILENAME EMAIL




34

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL




35

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL



36

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL

Let's make the email body prettier with ODS HTML


```
ods html3 body=mymail style=htmlblue;  
options nocenter;  
title;  
/* Start writing the email in a DATA _NULL_ step */  
data _null_;  
  file print;  
  put "Here is the information you asked for.";  
  put "Stay SASy!";  
  put "Mark";  
run;  
/* sql query results */  
...  
ods _all_ close;  
ods html;
```

37

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL



Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL

Let's take better control of the contents



```
ods html5 body=mymail style=htmlblue;  
options nocenter;  
title;  
proc odstext;  
  p 'Emailing from SAS is fun!' ;  
  p 'Here are a few reasons:' ;  
  list;  
    item 'SAS types better than I do';  
    item 'Everybody loves computer-generated email!';  
  end;  
  p 'Stay SASy!';  
  p 'Mark';  
run;  
ods html5 close;
```

39

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL



40

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL

Let's add reports!


```
ods html5 body=mymail style=journal3;
options nocenter;
title;
proc odstext;
  p 'Here is the information you requested.' ;
  p 'All the best,';
  p 'Mark';
run;
/* Insert query results */
PROC SQL;
title "US Automobile Stats by Make";
  SELECT Make
... additional report code ...
ods html5 close;
```

41

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL



Copyright © SAS Institute Inc. All rights reserved.





Emailing from SAS® with FILENAME EMAIL

Let's get fancy with an HTML body and a PDF attachment

First, create the PDF attachment file:

```
ods pdf body="%sysfunc(pathname(work))\Report.pdf" style=Journal3
pdftoc=1;
ods proclabel "Autos by Make";
title "US Automobile Stats by Make";
PROC SQL ;
    SELECT Make
          , COUNT(distinct Model) 'Models'
... additional report code ...
ods pdf close;
```

43

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL

In the FILENAME statement, add an attach= option:

```
filename mymail email "mark.jordan@sas.com"
    subject="US Automobile Stats" type="TEXT/HTML"
    attach="%sysfunc(pathname(work))\Report.pdf";
```

44

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL

Then, write the body of the email using ODS as before:

```
ods html5 body=mymail style=journal3;  
options nocenter;  
title;  
proc odstext;  
  p 'Here is the information you requested.' ;  
  p 'All the best,';  
  p 'Mark';  
run;  
  
ods html5 close;
```

45

Copyright © SAS Institute Inc. All rights reserved.



Emailing from SAS® with FILENAME EMAIL

US Automobile Stats - Message (HTML)

File Message Adobe PDF Tell me what you want to do

Delete Archive Reply Reply All Forward Quick Steps Move Tags Editing Zoom

Mark Jordan • Mark Jordan 1 8:14 AM

US Automobile Stats

If there are problems with how this message is displayed, click here to view

Report.pdf 216 KB

Here is the information you requested.

All the best,

Mark

Report.pdf - Adobe Acrobat Standard DC

Home Tools Report.pdf Report.pdf 1 / 4 75.7% Share

Bookmarks

- > Autos by Make
- > SUVs
- > Autos by Make and Type

US Automobile Stats by Make

Make	Models	Type	Overall	Average	Average
			MPG	MPG	Weight
Buick	9	2	23.3	214.4	3758
Cadillac	8	4	19.8	296.1	4364
Chevrolet	27	5	23.3	212.8	3640
Chrysler	15	3	23.6	201.1	3534
Dodge	13	4	22.8	239.7	3639
Ford	23	5	22.5	197.9	3749
GM	8	3	17.5	253.1	4731
Hummer	1	1	11.0	316.0	6400
Jeep	3	1	19.3	178.3	3730
Lincoln	9	2	20.1	262.3	4324
Mercury	9	3	21.1	210.7	3908
Oldsmobile	3	1	25.0	165.0	3326
Pontiac	11	4	24.2	195.9	3441
Saturn	8	3	28.8	145.6	2908

46

Copyright © SAS Institute Inc. All rights reserved.





Sending Email from SAS

This demonstration illustrates sending email from SAS.

`FWF07-sas_email_options.sas`



Copyright © SAS Institute Inc. All rights reserved.

Wrap Session

The SAS FILENAME statement is a virtual “Swiss army knife” for the SAS programmer. With FILENAME you can:

- Read data directly from a file on an FTP site without downloading it first (FILENAME FTP)
- Read data directly from a Web site without downloading it first (FILENAME URL)
- Read and write to ZIP file members without first unzipping the data (FILENAME ZIP)
- Send Email from a SAS program (FILENAME EMAIL)
- If you want to download materials from the internet, PROC HTTP makes it fast and easy.

48

Copyright © SAS Institute Inc. All rights reserved.



References

- [SAS® 9.4 Global Statements: Reference](#)
- [Jedi SAS Tricks: Email from the Front - Part 1](#)
- [Jedi SAS Tricks: Email from the Front - Part 2](#)

49


Copyright © SAS Institute Inc. All rights reserved.



Questions?



Email: Mark.Jordan@sas.com
Twitter: [@SASJedi](https://twitter.com/SASJedi)
Blog: <http://sasjedi.tips>
Author Page: <http://support.sas.com/jordan>



Copyright © SAS Institute Inc. All rights reserved.